
Vežba 3: Osnove SPICE simulatora 
111. Šta je SPICE program i čemu služi? 

SPICE je program za simulaciju elektronskih kola. 

112. Šta mora da bude prva linija SPICE netliste? 

Asterisk (*). 

113. Kojim simbolom počinju nazivi direktiva (komandi) u SPICE-u? 

Tačka (.). 

114. Koji numerički karakter je rezervisan za modelovanje čvora mase u SPICE-u? 

Nula (0). 

115. U tabeli ispod, pored naziva elementa kola upisati odgovarajući slovni simbol kojim se element 
opisuje u SPICE-u. 

Naziv elementa Opis u SPICE-u 
Bipolarni trazistor     Q 
Otpornik                     R 
MOS-FET trazistor    M 
Kondezator               C 
Dioda                         D 
Kalem                 L 
Naponski generator             V 
Strujni generator        I 
Naponski generator kontrolisan naponom E 
Strujni generator kontrolisan naponom G 
Naponski generator kontrolisan strujom H 
Strujni generator kontrolisan strujom F 

 
116. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Operating Point analize. 

Vrednosti elemenata kola su: I1=1mADC, V1=2V, V2=350mVDC, R1=10kΩ, R2=50Ω, R3=0.5kΩ. 

117. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Alternate Current analize. 

Analizirati kolo u opsegu od 100Hz do 100MHz. Opseg frekvencija zadati u logaritamskoj razmeri sa 
50 tačaka po dekadi. Vrednosti elemenata kola su: Rr=1kΩ, Lr=100mH, Cr=22nF, V1=0VDC (1VAC). 

 

 

118. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Direct Current sweep analize. 
Analizirati kolo za opseg napona V1 od -10V do 10V sa korakom od 0.01V. Naziv modela diode je 
DMOD i ima sledeće parametre: Is=1e-14 Vj=0.75 BV=6.8 M=0.5 Rs=5. Vrednosti elemenata kola 
su: V1=1VDC, R1=1kΩ. 

 

 

119. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Direct Current sweep analize. 
Analizirati kolo za opseg napona V1 od 0V do 5V sa korakom 0.01V i za listu struja I1 10μA, 20μA, 
30μA i 40μA. Naziv modela tranzistora je NBJT i ima sledeće parametre: Is=1e-15 Vje=0.7 Bf=120 
Vaf=150. Vrednosti elemenata kola su: I1=10uADC, V1=1VDC. 

* Primer 06 
I1 a 0 1m 
R1 a 0 10k 
V1 a b 2 
R3 b 0 0.5k 
R2 b c 50 
V2 0 c 350m 
.op 
.end 

* Primer 07 
Rr a b 1k 
Lr b c 100m 
Cr c 0 22n 
V1 a 0 0 AC 1 
.ac dec 50 100 100MEG 
.end 

* Primer 08 
V1 a 0 1 
R1 a b 1k 
D1 b 0 DMOD 
.model DMOD D (Is=1e-14 Vj=0.75 BV=6.8 M=0.5 Rs=5) 
.dc V1 -10 10 0.01 
.end 

120. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Direct Current sweep analize. 

Analizirati kolo za opseg napona V1 od 0V do 5V sa korakom 0.01V i za listu napona V2, 0.5V, 0.75V, 
1V, 1.25V. Naziv modela tranzistora je NFET sa sledećim parametrima: Kp=100e-6 Vto=0.4 
Lambda=0.05. Elemenati kola su: V1=1VDC, V2=1VDC. 

 

 

Vežba 4: Simulacija usmerača u SPICE-u 
121. Kojim elementima kola se modeluju primar i sekundar idealnog transformatora u SPICE-u? 

Spregnutim kalemovima. 

122. Za koju vrstu usmeravanja napona se koristi transformator sa srednjim izvodom na sekundaru? 

Dvostrano usmeravanje. 

123. Koliki odnos induktivnosti spregnutih kalemova treba odabrati da bi se u SPICE-u modelovao 
idealni transformator sa odnosom transformacije 1:16? 

* Primer 09 
I1 0 b 10u 
V1 c 0 1 
Q1 c b 0 NBJT 
.model NBJT NPN (Is=1e-15 Vje=0.7 Bf=120 Vaf=150) 
.dc V1 0 5 0.01 I1 list 10u 20u 30u 40u 
.end 

* Primer 10 
V1 d 0 1 
V2 g 0 1 
M1 d g 0 0 NFET 
.model NFET NMOS (Kp=100u Vto=0.4 Lambda=0.05) 
.dc V1 0 5 0.01 V2 list 0.5 0.75 1 1.25 
.end 

124. Koliki je maksimalni očekivani napon na izlazu jednostranog usmerača sa jednom diodom čiji je 
parametar VJ=0.6V, ako je maksimalni napon na izlazu sekundara transformatora 12V? 

𝑉𝑉𝑚𝑚′′ = 12𝑉𝑉 ⇒ 𝑉𝑉𝑚𝑚𝑚𝑚𝑚𝑚 = 𝑉𝑉𝑚𝑚′′ − 𝑉𝑉𝑗𝑗 = 11.4𝑉𝑉 

125. Kojim elementima se modeluju najjednostavnija kola za ograničavanje napona (clampers) u SPICE-
u?  

Diode i otpornici. 

126. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Transient analize. Simulacija kola 
treba da obuhvati 4 periode pobudnog signala. Naziv modela diode je 1N4001 i definisan je u fajlu 
diode.mod. Elementi kola su: R1=0.1Ω, RL=1kΩ, CF=10μF, L1=10mH, L2=10μH, K=1, 
V1=(310V)sin(2π(50Hz)t).  

 

 
127. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Transient i Direct Current sweep 

analize. Transient analiza treba da obuhvati 2 periode ulaznog pobudnog signala. Za Direct Current 
analizu zadati opseg napona Vi od -10V do 10V sa korakom od 0.01V. Naziv modela dioda je 1N914 
i definisan je u fajlu standard.dio. Elementi kola su: R1=1kΩ, V1=2V, V2=3V, Vi=1VDC + 
(5V)sin(2π(1kHz)t). 

 

* Primer 16 
V1 a 0 sin(0 310 50) 
R1 a b 0.1 
L1 b 0 10m 
L2 c 0 10u 
K  L1 L2 1 
D1 c d 1N4001 
RL d 0 1k 
CF d 0 10u 
.lib diode.mod 
.tran 80m 
.end 

128. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Transient i Direct Current sweep 

analize. Transient analiza treba da obuhvati 2 periode ulaznog pobudnog signala. Za Direct Current 
analizu zadati opseg ulaznog napona od -10V do 10V sa korakom od 0.01V. Naziv modela dioda je 
BZX84C6V2L i definisan je u fajlu standard.dio. Elementi kola su: R1=1kΩ, Vi=(12V)sin(2π(1kHz)t). 

 

 
129. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Transient analize. Transient 

analiza treba da traje jednu periodu ulaznog signala koji se najsporije menja. Naziv modela dioda je 
DMOD i ima zadat parametar Vj=0.75. Otpornost u kolu je: R1=1kΩ. Pobudne signale modelovati 
SPICE pulsnim generatorima koji imaju sledeće parametre:  

Va = {Vinit=0, Von=5, Tdelay=1m, Trise=1n, Tfall=1n, Ton=1m, Tperiod=2m} 

Vb = {Vinit=0, Von=5, Tdelay=2m, Trise=1n, Tfall=1n, Ton=2m, Tperiod=4m}   

* Primer 17 
Vi a 0 sin(1 5 1k) 
R1 a b 1k 
D1 c b 1N914 
V1 0 c 2 
D2 b d 1N914 
V2 d 0 3 
.lib standard.dio 
.tran 2m 
.dc Vi -10 10 0.01 
.end 

* Primer 18 
Vi a 0 sin(0 12 1k) 
R1 a b 1k 
D1 c b BZX84C6V2L 
D2 c 0 BZX84C6V2L 
.lib standard.dio 
.tran 2m 
.dc Vi -10 10 0.01 
.end 

130. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Transient analize. Transient 

analiza treba da traje jednu periodu ulaznog signala. Naziv modela dioda je 1N4148 i definisan je u 
fajlu standard.dio. Elementi kola su: R1=10kΩ, C1=10μF i Vi=1VDC + (10V)sin(2π(0.5kHz)t).  

 

 

Vežba 5: Simulacija pojačavača u SPICE-u 
131. Kojim tipom kontrolisanog generatora se modeluje naponski pojačavač u SPICE-u? 

Naponski generator kontrolisan naponom (E). 

132. Nabrojati najmanje četiri SPICE analize koje se obavezno koriste prilikom verifikacije i 
projektovanja pojačavača. 

.op, .tran, .ac, .noise 

* Primer 19 
Va a 0 pulse(0 5 1m 1n 1n 1m 2m) 
Vb b 0 pulse(0 5 2m 1n 1n 2m 4m) 
R1 c 0 1k 
D1 a c DMOD 
D2 b c DMOD 
.model DMOD D (Vj=0.75) 
.tran 4m 
.end 

* Primer 20 
Vi a 0 sin(1 10 0.5k) 
C1 a b 10u 
R1 b 0 10k 
D1 0 b 1N4148 
.lib standard.dio 
.tran 2m 
.end 

133. Koja SPICE analiza se koristi za procenu prenosne funkcije, ulazne i izlazne otpornosti pojačavača 
za male signale? 

.tf 

134. Koja SPICE analiza se koristi za procenu ekvivalentnog napona šuma na ulazu i izlazu pojačavača? 

.noise 

135. Koja SPICE analiza se koristi za procenu propusnog opsega pojačavača? 

.ac 

136. Napisati SPICE netlistu za model naponskog pojačavača sa sledećim parametrima: Rulp=10MΩ, 
Rizp=10Ω, Ao=100. Model enkapsulirati u pod-kolo pod nazivom VAMP. Usvojiti sledeći redosled 
navođenja portova: inp, inm, outp, outm. Pod-kolo VAMP instancirati i pobuditi generatorom, 
Vu=10mVDC, koji ima unutrašnju otpornost, Ru=10kΩ. Otpornost potrošača je, Rp=100Ω. Zadati 
Transfer Function analizu. 

 

 
137. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Transfer function analize. Izlaz 

pojačavača je na emitoru tranzistora, a ulaz pojačavača je pobuđen generatorom Vi. Naziv modela 
tranzistora je 2N2222 i definisan je u fajlu standard.bjt. Elementi kola su: R1=150kΩ, R2=300Ω, 
Vi=0.8VDC, VCC=5VDC. 

 

* Primer 26 
.subckt VAMP inp inm outp outm 
Rulp inp  inm  10MEG 
Eamp outi outm inp inm 100 
Rizp outi outp 10 
.ends 
Vu a 0 10m 
Ru a b 10k 
Xa b 0 c 0 VAMP 
Rp c 0 100 
.tf V(c) Vu 
.end 

138. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Alternate Current analize. 

Analizirati kolo u opsegu od 1Hz do 100MHz. Opseg frekvencija zadati u logaritamskoj razmeri sa 2 
tačke po oktavi. Naziv modela tranzistora je BCW60A i definisan je u fajlu standard.bjt. Elementi 
kola su: Ru=100Ω, RB=100kΩ, RC=450Ω, Rp=10kΩ, CS1=CS1=1μF, VCC=12VDC, Vi=0VDC (1VAC).  

 

 
139. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Transient analize. Analiza treba 

da obuhvati 5 perioda ulaznog signala. Naziv modela tranzistora je BSP89 i definisan je u fajlu 
standard.mos. Elementi kola su: RG1=100kΩ, RG2=70kΩ, RD=650Ω, RS=100Ω, Ru=50Ω, Rp=5kΩ, 
CS1=CS2=1μF, CS3=150nF, VDD=5VDC i Vi=1VDC + (10mV)cos(2π(10kHz)t).  

* Primer 27 
Vi  b 0 0.8 
R1  a b 150k 
Q1  c b e 2N2222 
R2  e 0 300 
Vcc a 0 5 
.lib standard.bjt 
.tf V(e) Vi 
.end 

* Primer 28 
Vi  1 0 AC 1 
Ru  1 2 100 
Cs1 2 3 1u 
Rb  3 6 100k 
Q1  4 3 0 BCW60A 
Rc  4 6 450 
Cs2 4 5 1u 
Rp  5 0 10k 
Vcc 6 0 12 
.lib standard.bjt 
.ac oct 2 1 100MEG 
.end 

140. Za kolo sa slike napisati SPICE netlistu sa komandom za pokretanje Direct Current sweep i Noise 

analize. Za Direct Curent analizu zadati promenu napona Vi od -10V do 10V sa korakom od 0.01V. 
Noise analizu zadati u opsegu od 10Hz do 10MHz sa 20 tačaka po dekadi. Izlazni napon pojačavača 
je napon čvora 5. Operacioni pojačavač je opisan kao pod-kolo pod nazivom OP747 čiji se opis nalazi 
u fajlu ADI.lib. Elementi kola su: R1=1kΩ, R2=3kΩ, R3=5kΩ, R4=10kΩ, Rp=100Ω, Cp=270nF, 
VCC=VEE=5VDC. 

 

* Primer 29 
Vi  1 0 sin(1 10m 10k 0 0 90) 
Ru  1 2 50 
Cs1 2 3 1u 
Rg1 3 7 100k 
Rg2 3 0 70k 
M1  5 3 4 4 BSP89 
Rd  5 7 650 
Rs  4 0 100 
Cs3 4 0 150n 
Cs2 5 6 1u 
Rp  6 0 5k 
Vdd 7 0 5 
.lib standard.mos 
.tran 500u 
.end 



Uvod u Elektroniku, pitanja vezana za SPICE simulator 

 
 

Vežba 6: Simulacija oscilatora u SPICE-u 
141. Navesti tri osnovna elementa kojima se modeluju LC oscilatori u SPICE-u? 

Kalem, kondenzator, pojačavač (tranzistor, operacioni pojačavač). 

142. Navesti tri osnovna elementa kojima se modeluju RC oscilatori u SPICE-u? 

Kondenzator, otpornik, pojačavač (tranzistor, operacioni pojačavač). 

143. Kojim elementima kola se obično ograničava izlazni napon oscilatora? 

Diodama. 

144. Koja je osnovna SPICE analiza koja se koristi prilikom verifikacije i projektovanja kola oscilatora? 

.tran 

145. Nakon SPICE simulacije i obrade izlaznog  signala oscilatora dobijen je spektar prikazan na slici. Na 
osnovu slike spektra proceniti osnovnu frekvenciju oscilovanja oscilatora. 

 
𝑓𝑓0 ≈ 12𝑘𝑘𝑘𝑘𝑘𝑘 

146. Za kolo Colpitts sa slike napisati SPICE netlistu sa komandom za pokretanje Transient analize. Analiza 
treba da obuhvati najmanje 500 perioda signala oscilatora. Naziv modela tranzistora je BCW60A i 

* Primer 30 
Vi  1 0 DC 0 
R1  2 0 1k 
R2  2 3 3k 
R3  3 4 5k 
R4  4 5 10k 
Rp  5 0 100 
Cp  5 0 270n 
X1  1 2 6 7 3 OP747 
X2  0 4 6 7 5 OP747 
Vcc 6 0 5 
Vee 0 7 5 
.lib ADI.lib 
.dc Vi -10 10 0.01 
.noise V(5) Vi dec 20 10 10MEG 
.end 

definisan je u fajlu standard.bjt. Elementi kola su: Lr=10mH, C1=240nF, C2=24nF, CS=1μF, R1=10kΩ, 
R2=500Ω i VCC=5V. 

 

𝐶𝐶𝑟𝑟 =
1

1
𝐶𝐶1

+ 1
𝐶𝐶2

= 21.82𝑛𝑛𝑛𝑛, 𝑓𝑓𝑜𝑜 =
1

2𝜋𝜋�𝐿𝐿𝑟𝑟𝐶𝐶𝑟𝑟
= 10.78𝑘𝑘𝑘𝑘𝑘𝑘,𝑇𝑇𝑠𝑠𝑠𝑠𝑠𝑠 ≥ �

500
𝑓𝑓𝑜𝑜[𝑘𝑘𝑘𝑘𝑘𝑘]

⋅ 103�𝑚𝑚𝑚𝑚 = 47𝑚𝑚𝑚𝑚 

 
147. Za kolo Colpitts oscilatora sa slike napisati SPICE netlistu sa komandom za pokretanje Transient 

analize. Analiza treba da obuhvati najmanje 300 perioda signala oscilatora. Naziv modela tranzistora 
je BSP89 i definisan je u fajlu standard.mos. Elementi kola su: Lr=10mH, C1=500nF, C2=50nF, 
CS1=0.1μF, CS2=1μF, RG1=100kΩ, RG2=70kΩ, RD=650Ω, RS=500Ω i VDD=5V. 

 

𝐶𝐶𝑟𝑟 =
1

1
𝐶𝐶1

+ 1
𝐶𝐶2

= 45.46𝑛𝑛𝑛𝑛, 𝑓𝑓𝑜𝑜 =
1

2𝜋𝜋�𝐿𝐿𝑟𝑟𝐶𝐶𝑟𝑟
= 7.465𝑘𝑘𝑘𝑘𝑘𝑘,𝑇𝑇𝑠𝑠𝑠𝑠𝑠𝑠 ≥ �

300
𝑓𝑓𝑜𝑜[𝑘𝑘𝑘𝑘𝑘𝑘]

⋅ 103�𝑚𝑚𝑚𝑚 = 41𝑚𝑚𝑚𝑚 

* Primer 36 
Q1  c b e BCW60A 
R1  b a 10k 
R2  e 0 500 
C1  c e 240n 
C2  e 0 24n 
Lr  c a 10m 
Cs  b 0 1u 
Vcc a 0 5 
.lib standard.bjt 
.tran 48m 
.end 

148. Za kolo Colpitts oscilatora sa slike napisati SPICE netlistu sa komandom za pokretanje Transient 

analize. Analiza treba da obuhvati 800 perioda signala oscilatora. Naziv modela tranzistora je 
BCW60A i definisan je u fajlu standard.bjt. Elementi kola su: LC=300mH, Lr=10mH, C1=40nF, 
C2=400nF, CS1=0.1μF, CS2=10nF, RB1=10kΩ, RB2=4.7kΩ, RE=1.5kΩ i VCC=5V. 

 

𝐶𝐶𝑟𝑟 =
1

1
𝐶𝐶1

+ 1
𝐶𝐶2

= 36.36𝑛𝑛𝑛𝑛, 𝑓𝑓𝑜𝑜 =
1

2𝜋𝜋�𝐿𝐿𝑟𝑟𝐶𝐶𝑟𝑟
= 8.346𝑘𝑘𝑘𝑘𝑘𝑘,𝑇𝑇𝑠𝑠𝑠𝑠𝑠𝑠 ≥ �

800
𝑓𝑓𝑜𝑜[𝑘𝑘𝑘𝑘𝑘𝑘]

⋅ 103�𝑚𝑚𝑚𝑚 = 96𝑚𝑚𝑚𝑚 

 

* Primer 37 
Lr  1 2 10m 
C1  1 0 500n 
C2  2 0 50n 
Cs1 2 3 0.1u 
Cs2 4 0 1u 
Rg1 3 5 100k 
Rg2 3 0 70k 
Rd  1 5 650 
Rs  4 0 500 
M1  1 3 4 4 BSP89 
Vdd 5 0 5 
.lib standard.mos 
.tran 42m 
.end 

* Primer 38 
Lc  1 5 300m 
Lr  1 2 10m 
C1  1 0 40n 
C2  2 0 400n 
Cs1 2 3 0.1u 
Cs2 4 0 10n 
Rb1 3 5 10k 
Rb2 3 0 4.7k 
Re  4 0 1.5k 
Q1  1 3 4 BCW60A 
Vcc 5 0 5 
.lib standard.bjt 
.tran 97m 
.end 

149. Za kolo oscilatora sa Wien mostom prikazano na slici napisati SPICE netlistu sa komandom za 
pokretanje Transient analize. Analiza treba da obuhvati 150 perioda signala oscilatora. Operacioni 
pojačavač je opisan kao pod-kolo pod nazivom OP747 čiji se opis nalazi u fajlu ADI.lib. Naziv modela 
dioda je 1N914 i definisan je u fajlu standard.dio. Elementi kola su: R1=1kΩ, R2=2.5kΩ, R3= 
R4=R=10kΩ, C3=C4=C=10nF i VCC=VEE=5V. 

 

𝑓𝑓𝑜𝑜 =
1

2𝜋𝜋𝜋𝜋𝜋𝜋
= 1.591𝑘𝑘𝑘𝑘𝑘𝑘,𝑇𝑇𝑠𝑠𝑠𝑠𝑠𝑠 ≥ �

150
𝑓𝑓𝑜𝑜[𝑘𝑘𝑘𝑘𝑘𝑘]

⋅ 103�𝑚𝑚𝑚𝑚 = 95𝑚𝑚𝑚𝑚 

 
150. Za kolo Phase-shift oscilatora prikazano na slici napisati SPICE netlistu sa komandom za pokretanje 

Transient analize. Analiza treba da obuhvati 1000 perioda signala oscilatora. Operacioni pojačavač 
je opisan kao pod-kolo pod nazivom OP747 čiji se opis nalazi u fajlu ADI.lib. Naziv modela dioda je 
1N914 i definisan je u fajlu standard.dio. Elementi kola su: R1=R2=R3=R=1kΩ, R4=40kΩ, 
C1=C2=C3=C=10nF i VCC=VEE=5V. 

* Primer 39 
R1  2 0 1k 
R2  2 3 2.5k 
R3  3 4 10k 
R4  1 0 10k 
C3  4 1 10n 
C4  1 0 10n 
D1  3 2 1N914 
D2  2 3 1N914 
X1  1 2 5 6 3 OP747 
Vcc 5 0 5 
Vee 0 6 5 
.lib ADI.lib 
.lib standard.dio 
.tran 96m 
.end 

𝑓𝑓𝑜𝑜 =
1

2𝜋𝜋𝜋𝜋𝜋𝜋√6
= 6.497𝑘𝑘𝑘𝑘𝑘𝑘,𝑇𝑇𝑠𝑠𝑠𝑠𝑠𝑠 ≥ �

1000
𝑓𝑓𝑜𝑜[𝑘𝑘𝑘𝑘𝑘𝑘]

⋅ 103�𝑚𝑚𝑚𝑚 = 154𝑚𝑚𝑚𝑚 

 

Vežba 7: Simulacija osnovnih logičkih kola u SPICE-u 
176. Koliko tranzistora je potrebno za realizaciju invertora u CMOS tehnologiji? 

Dva. 
177. Koliko tranzistora je potrebno za realizaciju logičke funkcije sa dva ulaza i jednim izlazom u CMOS 

tehnologiji? 
Četiri. 

178. Koji parametar specijalne funkcije u LTSpice-u određuje nivo logičke nule? 
Vlow. 

179. Koji parametar specijalne funkcije u LTSpice-u određuje nivo logičke jedinice? 
Vhigh. 

180. Kojim logičkim kolima se opisuju potpuni i polu-sabirač? 
XOR, AND i OR. 

181. Napisati SPICE netlistu za kolo statičkog CMOS bafera prikazano na slici. Naziv modela PMOS/NMOS 
tranzistora je NM/PM i definisan je u fajlu 180nm_bulk.txt. Dimezije NMOS tranzistora su 

* Primer 40 
R1  1 0 1k 
R2  2 0 1k 
R3  3 4 1k 
R4  4 5 40k 
R5  5 6 500 
C1  1 5 10n 
C2  1 2 10n 
C3  2 3 10n 
D1  6 0 1N914 
D2  0 6 1N914 
X1  0 4 7 8 5 OP747 
Vcc 7 0 5 
Vee 0 8 5 
.lib ADI.lib 
.lib standard.dio 
.tran 155m 
.end 

Wn=320nm i Ln=180nm. Dimezije PMOS tranzistora su Wp=620nm i Lp=180nm. Model enkapsulirati 
u pod-kolo pod nazivom BUF1x1. Usvojiti sledeći redosled navođenja portova: a, dd, ss i y. Pod-kolo 
instancirati i pobuditi naponskim generatorom koji ima sledeće parametre: 
Va = {Vinit=0, Von=1.8, Tdelay=0.5u, Trise=1n, Tfall=1n, Ton=0.498n, Tperiod=1u} 
Instacirano pod-kolo analizirati u vremenskom domenu. Transient analiza treba da obuhvati jednu 
periodu pobudnog signala. Nivoi logičke nule i jedinice su VSS=0V i VDD=1.8V, respektivno. Kolo 
opteretiti kapacitivnim opterećenjem od CL=1pF.    

 

 
182. Napisati SPICE netlistu za kolo statičkog CMOS NOR gejta prikazano na slici. Naziv modela 

PMOS/NMOS tranzistora je NM/PM i definisan je u fajlu 180nm_bulk.txt. Dimenzije svih NMOS 
tranzistora su Wn=320nm i Ln=180nm. Dimezije PMOS tranzistora su Wp=1.28um i Lp=180nm. Model 
enkapsulirati u pod-kolo pod nazivom NOR2x1. Usvojiti sledeći redosled navođenja portova: A, DD, 
SS i Y. Pod-kolo instancirati i pobuditi naponskim generatorima koji imaju sledeće parametre: 
Va = {Vinit=0, Von=1.8, Tdelay=0.5u, Trise=1n, Tfall=1n, Ton=0.498n, Tperiod=1u} 
Vb = {Vinit=0, Von=1.8, Tdelay=1u, Trise=1n, Tfall=1n, Ton=1.998u, Tperiod=2u}  
Instacirano pod-kolo analizirati u vremenskom domenu. Transient analiza treba da obuhvati jednu 
periodu pobudnog signala koji se sporije menja. Nivoi logičke nule i jedinice su VSS=0V i VDD=1.8V, 
respektivno. Kolo opteretiti kapacitivnim opterećenjem od CL=1pF.   

 

* Primer 46 
.subckt BUF1x1 a dd ss y 
Mn1 g a ss ss NM L=180n W=320n 
Mn2 y g ss ss NM L=180n W=320n 
Mp1 g a dd dd PM L=180n W=640n 
Mp2 y g dd dd PM L=180n W=640n 
.ends 
Vdd dd 0 1.8 
Vss ss 0 0 
Va a 0 pulse(0 1.8 0.5u 1n 1n 0.498u 1u) 
X1 a dd ss y BUF1x1 
CL y 0 1p 
.lib 180nm_bulk.txt 
.tran 1u 
.end 

183. Napisati SPICE netlistu za kolo multipleksera 2 u 1 prikazano na slici. Za opis gejtova koristiti 
specijalne funkcije u LTspice-u. Model enkapsulirati u pod-kolo pod nazivom MUX2x1. Usvojiti 
sledeći redosled navođenja portova: A, B, SEL i Y. Pod-kolo instancirati i pobuditi naponskim 
generatorima koji imaju sledeće parametre: 
Vsel = {Vinit=0, Von=1.8, Tdelay=0.5u, Trise=1n, Tfall=1n, Ton=0.498n, Tperiod=1u} 
Va = {1.8VDC}, Vb={0VDC}  
Instacirano pod-kolo analizirati u vremenskom domenu. Transient analiza treba da obuhvati dve 
periode selektorskog signala, SEL. Nivoi logičke nule i jedinice su VSS=0V i VDD=1.8V, respektivno. 
Kolo opteretiti kapacitivnim opterećenjem od CL=1pF.  
 

 
 

 
184. Napisati SPICE netlistu za kolo potpunog sabirača prikazano na slici. Za opis gejtova koristiti 

specijalne funkcije u LTspice-u. Model enkapsulirati u pod-kolo pod nazivom FA. Usvojiti sledeći 

* Primer 47 
.subckt NOR2x1 a b dd ss y 
Mn1 y a ss ss NM L=180n W=320n 
Mn2 y b ss ss NM L=180n W=320n 
Mp1 d a dd dd PM L=180n W=1.28u 
Mp2 y b d  dd PM L=180n W=1.28u 
.ends 
Vdd dd 0 1.8 
Vss ss 0 0 
Va a 0 pulse(0 1.8 0.5u 1n 1n 0.498u 1u) 
Vb b 0 pulse(0 1.8 1u   1n 1n 1.998u 2u) 
X1 a b dd ss y NOR2x1 
CL y 0 1p 
.lib 180nm_bulk.txt 
.tran 2u 
.end 

* Primer 48 
.subckt MUX2x1 a b sel y 
A1 sel 0   0 0 0 1 0 0 BUF Vhigh=1.8 
A2 a   sel 0 0 0 0 2 0 AND Vhigh=1.8 
A3 1   b   0 0 0 0 3 0 AND Vhigh=1.8 
A4 2   3   0 0 0 0 y 0 OR Vhigh=1.8 
.ends 
Va   a   0 1.8 
Vb   b   0 0 
Vsel sel 0 pulse(0 1.8 0.5u 1n 1n 0.498u 1u) 
Xmux a   b sel y MUX2x1 
.tran 2u 
.end 

redosled navođenja portova: A, B, Ci, Co i S. Pod-kolo instancirati i pobuditi naponskim 
generatorima koji imaju sledeće parametre: 
Vci = {Vinit=0, Von=1.8, Tdelay=0.5u, Trise=1n, Tfall=1n, Ton=0.498n, Tperiod=1u} 
Vb = {Vinit=0, Von=1.8, Tdelay=1u, Trise=1n, Tfall=1n, Ton=0.998n, Tperiod=2u} 
Va = {Vinit=0, Von=1.8, Tdelay=2u, Trise=1n, Tfall=1n, Ton=1.998n, Tperiod=4u} 
Instacirano pod-kolo analizirati u vremenskom domenu. Transient analiza treba da obuhvati dve 
periode selektorskog signala. Nivoi logičke nule i jedinice su VSS=0V i VDD=1.8V, respektivno. Izlaze 
kolo opteretiti kapacitivnim opterećenjem od CL=1pF.  
 

 
 

 
 

185. Napisati SPICE netlistu za kolo trobitnog sabirača prikazano na slici. Model enkapsulirati u pod-kolo 
pod nazivom ADD3b. Usvojiti sledeći redosled navođenja portova: A2, A1, A0, B2, B1, B0, S2, S1, S0, 
Co. Opis pod-kola FA i HA se nalazi u fajlu logic.ckt. Redosled portova pod-kola FA je A, B, Ci, Co i S. 
Redosled portova pod-kola HA je A, B, Co i S. Pod-kolo CNT3b instancirati i pobuditi naponskim 
generatorima koji imaju sledeće parametre: 
Vb2 = {0VDC}, Vb1={1.8VDC}, Vb0={1.8VDC} 
Va2 = {0VDC}, Va1={1.8VDC}, Va0={0VDC} 
Zadati Operating Point analizu. Nivoi logičke nule i jedinice su VSS=0V i VDD=1.8V, respektivno. 
Izlaze kolo opteretiti kapacitivnim opterećenjem od CL=1pF.    

* Primer 49 
.subckt FA a b ci co s 
A1 a  b  0 0 0 0 1  0 XOR Vhigh=1.8 
A2 1  ci 0 0 0 0 s  0 XOR Vhigh=1.8 
A3 ci 1  0 0 0 0 2  0 AND Vhigh=1.8 
A4 a  b  0 0 0 0 3  0 AND Vhigh=1.8 
A5 2  3  0 0 0 0 co 0 OR  Vhigh=1.8 
.ends 
Vci ci 0 pulse(0 1.8 0.5u 1n 1n 0.498u 1u) 
Vb  b  0 pulse(0 1.8 1u 1n 1n 0.998u 2u) 
Va  a  0 pulse(0 1.8 2u 1n 1n 1.998u 4u) 
Xfa a  b ci co s FA 
CL1 co 0 1p 
CL2 s  0 1p 
.tran 4u 
.end 

* Primer 50 
.subckt ADD3b a2 a1 a0 b2 b1 b0 s2 s1 s0 co 
Xha0 a0 b0 1    s0 HA 
Xfa1 a1 b1 1 2  s1 FA 
Xfa2 a2 b2 2 co s2 FA 
.ends 
Vb2  b2 0 0 
Vb1  b1 0 1.8 
Vb0  b0 0 1.8 
Va2  a2 0 0 
Va1  a1 0 1.8 
Va0  a0 0 0 
Xadd a2 a1 a0 b2 b1 b0 s2 s1 s0 co ADD3b 
Cs2  s2 0 1p 
Cs1  s1 0 1p 
Cs0  s0 0 1p 
Cco  co 0 1p 
.lib logic.ckt 
.op 
.end 

Dodatak 
Sintaksa osnovnih SPICE komandi 

Opis SPICE sintaksa1 

Operating Point analiza .op 

Direct Current (DC) analiza .dc <Srcnam1> <Start> <Stop> <Incr>  
[+ <Srcnam2> list <Val1> [<Val2> ...]] 

Alternate Current (AC) analiza .ac <oct, dec, lin> <Nsteps> <StartFreq> <EndFreq> 

.ac list <FirstFreq> [<NextFreq> [<NextFreq> ...]] 

Transfer Function (TF) analiza .tf V(<node>[, <ref>]) <source> 

.tf I(<voltage source>) <source> 

Noise analiza .noise V(<outNode>[,<refNode>]) <Srcnam>  
+ <oct, dec, lin> <Nsteps> <StartFreq> <EndFreq> 

.noise V(<out>[,<ref>]) <src> list <FirstFreq> 
[ <NextFreq> [<NextFreq> ...]] 

Traisient analiza .tran <Tstep> <Tstop> [Tstart [dTmax]] [modifiers] 

.tran <Tstop> [modifiers] 

Uključivanje modela/pod-kola 
iz eksternog fajla 

.lib <fileName> 

.inc <fileName> 

Modelska kartica .model <ModelName> D|NMOS|PMOS|NPN|PNP|JFN|JFP ...  
+ [(<prameters>)] 

Opis pod-kola .subckt <SubCktName> <pin1> [<pin2> ...] 

<SubCktDescription> 

.ends 

Instaciranje primitiva/pod-
kola 

Rnnn|Lnnn|Cnnn|Dnnn|Mnnn|Qnnn|Jnnn|Vnnn|Innn|Xnnn  

+ |... <pin list>  

+ <ModelName>|<SubCktName> [<Properties>] 

Instaciranje specijalnih 
funkcija (logička kola)2 

Annn <in1> <in2> <in3> <in4> <in5>  

+ <notOut> <out> <ref> 

+ INV|BUF|AND|OR|XOR|... [extra parameters] 

Generator kontrolisan 
naponom (između čvorova 
ctrl+ i ctrl-). 

E|G <out+> <out-> <ctrl+> <ctrl-> 
+ <Gain|TransConductance> 

Genereator kontrolisan 
strujom (kroz generator Vnnn) 

F|H <out+> <out-> Vnnn <Gain|TransResistance> 

Spregnuti kalemovi3 Knnn L1 L2 [L3 ...] <Kcoeff> 

Sinusni generator sine(Voffset Vamp Freq Tdeay Damping Phase Ncycles) 

Pulsni generator pulse(Vinit Von Tdelay Trise Tfall Ton Tperiod 
Ncycles) 

Pice Wise Linear generator pwl(t1 v1 t2 v2 t3 v3...) 

Redosled navođenja pinova 
tranzistora i pod-kola 
operacionog pojačavača. 

BJT: <Collector> <Base> <Emitter> [<Substrate>] 

MOS: <Drain> <Gate> <Source> <Bulk> 

JFET: <Drain> <Gate> <Source> 

OPAMP: <In+> <In-> <Supply+> <Supply-> <Out> 
<SubcktName> 

1 [] – Opcioni argumenti, <> – Opis vrednosti/argumenta, | – Logičko „ili“, n – Dezignator (indeks).  
2 Model: INV, BUF, AND, OR, XOR, SCHMITT, SCHMTBUF, SCHMTINV, DFLOP, VARISTOR, MODULATE 
  Napomena: Pinove koji se ne koriste povezati na masu. 
3 Kcoeff – Koeficijent sprege (međuinduktivnosti). 

 

Frekvencija oscilovanja Colpitts LC oscilatora: 

𝑓𝑓𝑜𝑜 =
1

2𝜋𝜋�𝐿𝐿𝑟𝑟𝐶𝐶𝑟𝑟
,𝐶𝐶𝑟𝑟 =

1
1
𝐶𝐶1

+ 1
𝐶𝐶2

 

Frekvencija oscilovanja Wien RC oscilatora: 

𝑓𝑓𝑜𝑜 =
1

2𝜋𝜋𝜋𝜋𝜋𝜋
 

Frekvencija oscilovanja phse shift RC oscilatora: 

𝑓𝑓𝑜𝑜 =
1

2𝜋𝜋𝜋𝜋𝜋𝜋√6

